



Programming Using Visual Basic, Basic Course

Assignment 4

One and two dimensional arrays

Help and guidance

Cinema Booking System

Version 2

[Farid Naisan](#)

University Lecturer
Department of Computer Sciences
Malmö University, Sweden



Contents

Help on Assignment4: CBS Versions2 & 3 - Arrays.....	3
1. Objectives	3
Version 2a: One-dimensional array	3
2. Class diagram	3
3. SeatManager	4
4. The MainForm	8
Version 2b: Two-dimensional array	11
5. The SeatManager Class	11
6. The DisplayOptions enum	12
7. The MainForm Class	13
8. Working with Enums in conjunction with ComboBoxes and ListBoxes	14
9. Mapping the ListBox-items to rows and columns in the SeatManager class	15
10. Multidimensional arrays vs one-dimensional array of objects	18
Version 3: Array of objects	19
11. The MainForm Class	19
12. The Seat Class	20
13. The SeatManager Class	20

Help on Assignment4: CBS Versions2 & 3 - Arrays



1. Objectives

The objectives of this document are to give some help, hints and guidance to assist you accomplish the assignment. We begin with the part that is required for a passing grade. Some instructions will also be provided for those who are aiming at higher grade and would like to do the extra requirements.

Version 2a: One-dimensional array

2. Class diagram

The left part of the diagram shows a list of the methods that you might need to write. It also shows the instance variables needed. `m_numOfSeats` is a private constant that you can initiate to any number you wish, for ex 60 or 240 and this will determine the size of the array to be created in the `SeatManager` object.



3. SeatManager

3.1 Create a class **SeatManager** and save it as SeatManager.vb.

3.2 The information to be saved in the program is the customer's name and the price paid for the seat. Declare two one-dimensional arrays in this class. Two one-dimensional arrays, hereafter referred to as **m_nameList** and **m_priceList** can be used, one for storing the names of the customer, and one for the prices of the seats.

3.3 The seats are arranged in a number of rows and each row has an equal number of columns.

3.4 Fields:

- 3.4.1 **m_totNumOfSeats** is the number of seats available (maximum size of the arrays). The value of this **ReadOnly** variable is initialized through a constructor parameter. A read only type can only be assigned a value when it is initiated, either at declaration or in a constructor. The first alternative has the same effect as constant variable, while the latter allows the client objects (other objects that use this object) to provide an initial value. In this assignment we let MainForm set the value through the constructor call.
- 3.4.2 **m_nameList**: array indexed from 0 to **m_totNumOfSeats** -1 is used for storing the name of the customer for whom a seat is reserved. A vacant seat has the value **Nothing**.
- 3.4.3 **m_priceList**: array indexed from 0 to **m_totNumOfSeats** -1 is used for storing the prices paid for a seat; it has the value 0.0 if a seat is not reserved.
- 3.4.4 In the code example here, the arrays are declared but not created. Remember that arrays are objects and they must be created using the keyword **New**. There are at least two places you perform the declaration: (1) by initiation on the same lines the declaration:

```
Private m_nameList As String() = New String(m_totNumOfSeats - 1);
```

(2) in the constructor as shown in the code clip down here.

- 3.4.5 Write a constructor with one parameter to expect an initial value for the variable **m_totNumOfSeats**. Because the **m_totNumOfSeats** is declared **ReadOnly** in the code, it can only get a value in the constructor or at declaration. As we want the client objects (MainForm) to decide about the

```
Public Class SeatManager
    'Input variables
    'total number of seats, initialized in the constructor
    Private ReadOnly m_totNumOfSeats As Integer
    'array for storing customer names
    Private m_nameList As String() = Nothing
    'array for storing seat prices
    Private m_priceList As Double() = Nothing
```

```
Public Sub New(ByVal maxNumberOfSeats As Integer)
    'only time m_totNumOfSeats can be assigned a value
    m_totNumOfSeats = maxNumberOfSeats
    m_nameList = New String(m_totNumOfSeats) {}
    m_priceList = New Double(m_totNumOfSeats) {}
End Sub
```

size of the array, we choose the latter option, i.e. initiating `m_totNumOfSeats` in the constructor.

- 3.5 The table below lists the methods in the `SeatManager` class as listed in the class diagram presented earlier and describes them with some more details. You don't have to do the same methods if you prefer to redesign the solution. The same is true for the methods in the `MainForm`.

3.6 Methods	Description
<pre>''' <summary> ''' Constructor - do all your initializations here ''' </summary> ''' <param name="maxNumberOfSeats">Total number of seats</param> ''' <remarks></remarks> Public Sub New(ByVal maxNumberOfSeats) 'only time m_totNumOfSeats can be assigned a value m_totNumOfSeats = maxNumberOfSeats m_nameList = New String(m_totNumOfSeats - 1) {} m_priceList = New Double(m_totNumOfSeats - 1) {} End Sub</pre>	<p>Constructor with one parameter containing a value for the total number of seats. This value determines the size of the arrays.</p> <p>When MainFrame creates an object of this class, it also sends a value, for ex 240 to this class via this constructor. The value is then saved in the m_totNumOfSeats in the object, and the array is created with this size.</p>
<pre>''' <summary> ''' Check so the value of an index is within the array range, ''' i.e. index >=0 and index is less than m-totNumOfSeats. ''' </summary> ''' <param name="index"></param> ''' <returns></returns> ''' <remarks></remarks> Private Function CheckIndex(ByVal index As Integer)</pre>	<p>A help function used internally in the class to check the value of an index so it is not out of range. It returns true if the parameter index is ≥ 0 and $<$ the total number of seats. (The last position in an array is one less than the size of the array). The outer boundary of the array must be less than m_nameList.Length.</p> <p>The method needs not to be exposed to other objects and therefore it is declared as private.</p>
<pre>''' <summary> ''' Calculate the number of seats reserved. ''' </summary> ''' <returns>number of reserved seats</returns> Public Function GetNumReserved() As Integer</pre>	<p>Calculates the total number of reserved seats. Runs a loop through the array, calculates and returns the number of seats that is reserved. A seat is reserved if the value of the element in the m_nameList in a position is not null or empty. You can use the [String].IsNullOrEmpty method. The GetNumReserved method is called by the MainForm to update the output on the GUI.</p>



<pre>''' <summary> ''' Calculate the number of seats not reserved. ''' </summary> ''' <returns>Number of vacant seats</returns> Public Function GetNumVacant() As Integer</pre>	<p>As above, but now check so the value saved in an element in <code>m_nameList</code> is null or empty.</p> <p>This method is called by the MainForm to update output on the GUI.</p>
<pre>''' <summary> ''' Get total number of seats. MainForm already has this number. ''' But it's good to have this function anyway. ''' </summary> ''' <returns>The total number of seats</returns> Public Function GetNumOfSeats() As Integer Return m_totNumOfSeats End Function</pre>	<p>Return the m_totNumOfSeats as this is a private instance variable.</p> <p>MainFrame in this assignment sets this values but it is good to have the function, so you are prepared to have this data from the user as input in the future.</p>
<pre>''' <summary> ''' Adds a reservation. Save name in the namelist and the price in the pricelist ''' in the position= index ''' </summary> ''' <param name="index">Index of the array position.</param> ''' <param name="name">Name of the cinema customer</param> ''' <param name="price">Price paid for the seat.</param> ''' <returns>True if seat is successfully reserved, False if it is already ''' occupied</returns> Public Function ReserveSeat(ByVal name As String, ByVal price As Double, _ ByVal index As Integer) As Boolean</pre>	<p>This method is to be called whenever a new seat reservation is to be saved, i.e when the Reserve button is pressed in the MainForm.</p> <p>The method receives the required input through its argument list, and saves the data in the related arrays in the SeatManager object. The index comes from the MainForm's ListBox (SelectedIndex)</p>
<pre>''' <summary> ''' Cancel a reservation. Assign a value Nothing in the namelist, and 0.00 in ''' pricelist in the position = index ''' </summary> ''' <param name="index">Index for array position.</param> ''' <returns>true if seat was successfully canceled, false if the seat already ''' occupied.</returns> Public Function CancelSeat(ByVal index As Integer) As Boolean</pre>	<p>This method is to be called whenever a reservation for a seat is to be canceled.</p> <p>The SeatManager object marks the seat in the position = index as vacant by setting the value of the element in m_nameList to Nothing and the price in the m_priceList to 0.0d. The index comes from the MainForms ListBox (SelectedIndex).</p>



```
''' <summary>
''' Returns the status for a seat in a position = index
''' </summary>
''' <param name="index">Index of the array position</param>
''' <returns>A formatted string containing information about the seat,
''' customer name, price
''' and whether the seat is reserved or vacant.</returns>
Public Function GetSeatInfoAt(ByVal index As Integer) As String
```

This method can be used by **MainForm** to acquire data to display item by item in the ListBox.

This is a method that formats and returns an output string for a seat at the position number that is equal to index. The method can be called for every seat using a loop inside the class or in the **MainFrame**. The latter alternative is more object-oriented and is to prefer. The method below uses this pattern.

```
''' <summary>
''' This method prepares an array of strings with information about all seats.
''' Each element is a string formatted using the GetSeatInfo function.</
''' </summary>
''' <returns> </returns>
Public Function GetSeatInfoStrings(ByVal choice As DisplayOptions) As String()
    Dim count As Integer = GetNumOfSeats(choice)

    If (count <= 0) Then
        Return Nothing
    End If

    Dim strSeatInfoStrings As String() = New String(count - 1) {}

    Dim i As Integer = 0 'need for the strSeatInfo
    ' is the element corresponding with the index empty
    For index As Integer = 0 To m_totNumOfSeats - 1
        strSeatInfoStrings(index) = GetSeatInfoAt(index)
    Next
    Return strSeatInfoStrings
End Function
```

This method returns an array of strings, in which each element is a string that is formatted with the name of the customer and the price of the seat. This information is obtained from the name and price arrays.

MainForm can call this method and then it can use the ListBox's **AddRange** method to add and display the whole array to the ListBox – *fantastic!*

You have all the code for this method, but you are expected to take a moment and review all the lines to make sure that you understand the code fully.

4. The MainForm

The **MainForm** copied from the previous version (last assignment) needs to undergo some changes.

The code-clip given here at the right is to help you construct your **MainForm** object. The table that follows summarizes major changes and revisions required to program the features of your application.

4.1 Fields:

- 4.1.1 **m_NumOfSeats** is the number of seats. This value is passed to **m_seatMngr** object as a constructor parameter at the time the object is created.
- 4.1.2 **m_seatMngr** is an object of the **SeatManager** class and is used as a field by **MainForm** (aggregation) to hold data for all seats in the movie.

```
Public Class MainForm
    'Fields
    'Declare a constant for max number of seats in the cinema
    Private Const m_numOfSeats As Integer = 60

    'Declare a reference variable of the SeatManager type
    Private m_seatMngr As SeatManager

    '''Constructor is a special method that is automatically called
    '''when an instance of the class is created by using the keyword
    '''new. It is a good place for initializations and creation of
    '''the objects that are used as fields, e.g. m_seatMngr
    ...

    Public Sub New() 'constructor
        ' This call is required by the designer.
        InitializeComponent()

        ' Add any initialization after the InitializeComponent() call.
        m_seatMngr = New SeatManager(m_numOfSeats) 'Keyword New - Very important
        InitializeGUI()
    End Sub
```

4.2 Methods	Description
<pre>''' <summary> ''' Clear the input and output controls (if needed). ''' Do other initializations, for example select one of the radio- ''' buttons as default. ''' Create ''' </summary> ''' <remarks>This is to be called from the constructor, AFTER the ''' call to InitializeComponents.</remarks> Private Sub InitializeGUI()</pre>	<p>Initialize the input/out controls as in the last</p> <p>Call your method UpdateGUI, so the ListBox shows all seats, and updates other output data in the controls (Labels) designed for them, whenever a seat is reserved or a reservation is cancelled.</p> <p>It is usually a good idea to have a method of this type in your forms to call whenever a change in input/output requires an update of the GUI.</p>



<pre>''' <summary> ''' The user must highlight an item in the Listbox before a ''' reservation/cancelation can be performed. If an item in ''' the listbox is not highlighted, give an error msg to the user. ''' </summary> ''' <returns></returns> ''' <remarks></remarks> Private Function CheckSelectedIndex() As Boolean</pre>	<p>A utility function that checks whether an item on the ListBox is highlighted by the user (or from the program code). It returns true if an item is highlighted and false otherwise. It informs the user through a messagebox to select an item before a reservation or cancellation can take place.</p> <p>The value of SelectedIndex is the number equal to the row number (counted from 0) of the item highlighted in the ListBox. When no line in the ListBox is selected, the value of the SelectedIndex will be set to -1 automatically.</p>
<pre>Private Function ReadAndValidateInput(ByRef name As String, _ ByRef price As Double) As Boolean . Private Function ReadAndValidateName(ByRef name As String) As Boolean Private Function ReadAndValidatePrice(ByRef price As Double) As Boolean</pre>	<p>These methods are from the last assignment – no changes should be necessary</p>
<pre>''' <summary> ''' Event-handler method for the Click-event of the button. When the user ''' clicks the button, this method will be executed automatically. ''' If the Cancel RadioButton is checked, no need to read customer name ''' or seatPrice. ''' Call the method ReserveOrCancelSeat to process the reservation/cancellation ''' of a seat. ''' </summary> ''' <param name="sender"></param> ''' <param name="e"></param> ''' <remarks>Don't worry about the sender and the parameter e at this ''' time</remarks> ''' Private Sub btnOK_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) _ Handles btnOK.Click</pre>	<p>The logics for reserving or canceling a seat is placed in this method, held within another method (ReservOrCancelSeat()) that is called from here.</p>



<pre>''' <summary> ''' Reserve or cancel a seat ''' 1. Check the the user has highlighted a row on the listbox ''' If not, give a friendly message to user and return. ''' 2. If the Reserve option button is checked, ''' 2.a If the seat is already checkd, confirm with the ''' user to continue or return ''' 2.b If continue, call ReadAndValidateInput to read the ''' name and price from the textboxes ''' 2.c If reading is OK, call the m_seatMngr's Reserve method ''' to reserve the seat. ''' 3. Else if the Cancel option button is checked, ''' Call the m_seatMngr's CancelReservation method. ''' 4. Call the UpdateGUI method to update the output controls. ''' ''' </summary> ''' Private Sub ReserveOrCancelSeat()</pre>	<p>This method is an important one as it starts the main task of the program, i.e. to reserve or cancel a seat. The algorithm for this method is given as XML comments above the method signature. Follow the algorithm given at the left or apply your own.</p>
<pre>''' <summary> ''' Clear output controls (if needed). ''' Fill the listbox with info for varje seat. Each row in the ''' Listbox is to represent a seat. ''' Update also the labes with values for the num of reserved/vacant ''' seats. ''' </summary> ''' Private Sub UpdateGUI()</pre>	<p>Clears the ListBox and refills it with updated information for every seat taken from the object of the SeatManager, m_SeatMngr.</p> <p>Call the GetSeatInfoStrings() of the m_seatMngr to receive an array of strings. Add the array to the ListBox, using its AddRange method. Update also the output controls at the left side.</p>

Version 2b: Two-dimensional array

In this version seats are divided into a number of rows and columns. A seat is identified by a double index (row, col). For example, **m_nameMatrix(4, 3)** refers to the seat on row number 5 and column number 4 (seats are displayed from 1 as in the theatre, but in our arrays they are numbered from 0).

5. The SeatManager Class

5.1 Declare two two-dimensional arrays (m_nameMatrix and m_priceMatrix). The following code example may be of help:

```
Public Class SeatManager
    'files
    Private ReadOnly m_totNumOfCols As Integer = 12 'initialized through constructor
    Private ReadOnly m_totNumOfRows As Integer = 20 'initialized through constructor

    'the seat names stored in a two-dimensional array
    Private m_nameMatrix As String(,)

    ''' <summary>
    ''' Constructor which initates the size of the arrays
    ''' </summary>
    Public Sub New(ByVal totNumofRows As Integer, ByVal totNumOfCols As Integer)
```

5.2 Write a constructor with two parameters, one for input of total number of rows and one for input of the total number of columns.

5.3 The total number of rows and the total number of columns are set by the client objects (MainForm here) when calling the constructor as explained earlier.

5.4 Change the methods and do other necessary coding so your **SeatManager** class is now working with two-dimensional arrays instead of one-dimensional ones. If you wish to get some help, some code snippets are provided at the end of this document.

5.5 When working with two-dimensional arrays, it is quite practical to run a nested loop as below:

```
For row As Integer = 0 To m_totNumOfRows -1

    'You may have code for every row here
    For col As Integer = 0 To m_totNumOfCols -1

        'here comes code to be executed for
        'every column of the current row.
    Next
Next
```

- 5.5.1 A method that returns an index for an element at the position (row, col) that corresponds to a position in a one-dimensional presentation of the matrix. It requires a simple algorithm (solution given at the end of this document).
- 5.5.2 A method that returns an index to an element saved in the matrix at the position(row, col) that corresponds to a given index in a one-dimensional array (ListBox), (solution given at the end of this document).

6. The DisplayOptions enum

Write a Public Enum **DisplayOptions** to define a number of choices as in the figure. Save this enum in a separate file called DisplayOptions.vb. Enumerations are very practical to group named constants. An **Enum** is a type in C# and you instantiate them. The .NET Framework has a ready to use object called, `[Enum]` that operates on **Enums** and provides many useful services.

This enum can be used in the SeatManager as an option for calculating values or preparing information for each

```
Public Enum DisplayOptions
    AllSeats
    VacantSeats
    ReservedSeats
End Enum
```

choice (more info below).

7. The MainForm Class

7.1 To fill the ComboBox with members of the DisplayOption and set a default item, the following code can be used:

```
cmbDisplayOptions.Items.AddRange([Enum].GetNames(GetType(DisplayOptions)))  
cmbDisplayOptions.SelectedIndex = Convert.ToInt32(DisplayOptions.AllSeats)
```

7.2 It should however be mentioned that there is also another smooth way of filling the ComboBox through binding:

```
cmbDisplayOptions.DataSource = [Enum].GetNames(GetType(DisplayOptions))
```

The first item in the enum will be selected automatically in this alternative.

7.3 The GUI should now also manage the input for both rows and columns and **ListBox** should display these too as illustrated in the assignment description.

7.4 Write code in the **InitializeGUI** method to fill the **ComboBox** with items for the **DisplayOption**enum. Two-ways to do that:

7.5 Bring other necessary changes in the code file to get things work.

8. Working with Enums in conjunction with ComboBoxes and ListBoxes

8.1 To convert from the **SelectedIndex** (which returns an integer) to a value of an **Enum** type, you can use the following example:

```
DirectCast(cmbDisplayOptions.SelectedIndex, DisplayOptions)
```


and to convert from an Enum-element to an Integer you can write:

```
cmbDisplayOptions.SelectedIndex = Convert.ToInt32(DisplayOptions.AllSeats)
```


8.2 ListBoxes and ComboBoxes have many properties and methods that are common to both. The properties **Items.Add**, **Items.AddRange**, **SelectedIndex** and **SelectedItem** are some examples. The main difference between a ComboBox and ListBox is that a ComboBox has also a textbox part which can be edited by the user.

*ComboBox
EditableTextBox part*

8.3 The user can also select an option from the list. The text content is saved in the ComboBox's Text property (cmbDisplayOption.Text). If you don't code to work with the feature of the ComboBox, you should not allow the user to edit the textbox part of the ComboBox. To accomplish this, change the **DropDownStyle** of the ComboBox to **DropDownList** using the Properties window in the VS designer.



*Change to disallow use
of the textbox part.*



8.4 To fill the elements of an Enum to a ComboBox, you can do as follows:

```
//Add the DisplayOptions to the ComboBox  
cmbDisplayOptions.Items.AddRange(Enum.GetNames(GetType(DisplayOptions)))
```

8.5 The ComboBox's **SelectedIndex** gets the value of 0 or higher when an option in the list is selected by the user (just as with ListBoxes). When no entry is selected, the **SelectedIndex** has a value of -1. It is important to always check that the **SelectedIndex** is 0 or higher before using the index.

8.6 To select the first entry (option) in the ComboBox as default, set

```
cmbDisplayOptions.SelectedIndex = 0
```

9. Mapping the ListBox-items to rows and columns in the SeatManager class

- 9.1 As a programmer you can almost never avoid lists of data of the same type. Databases are of course one of the best solutions, but it is not always that you can work directly with databases
- 9.2 The hardest task in this version is perhaps to map an entry in the list box (where the items are a one-dimensional array) to an index (row, col) in your matrices. Items in the ListBox are indexed as a one-dimensional array from 0 to number of items -1. This information is available to your code as the value of the ListBox's SelectedIndex. (IstReservations.SelectedIndex = 7 in the run example below).

col → 0 1 . . 7 8 9

0
1
2
3

SelectedIndex = 7
Value to be saved:
m_nameList(0,7) = "Nisse"
m_priceList(0,7) = 67.0

Apu Movies & Theatre (V2)

Booking Input

☒ Reserve ☐ Cancel Resv.

Name: Nisse Price: 67

Reserve / Cancel

Output Data

Number of Seats: 240
Number of Reserved: 1
Number of Vacant: 239

Reservations

Show AllSeats

Row	Col	Status	Name	Price
1	1	Vacant	-	0,00 kr
1	2	Vacant	-	0,00 kr
1	3	Vacant	-	0,00 kr
1	4	Vacant	-	0,00 kr
1	5	Vacant	-	0,00 kr
1	6	Vacant	-	0,00 kr
1	7	Vacant	-	0,00 kr
1	8	Reserved	Nisse	67,00 kr
1	9	Vacant	-	0,00 kr
1	10	Vacant	-	0,00 kr
1	11	Vacant	-	0,00 kr
1	12	Vacant	-	0,00 kr

- 9.3 The items in the ListBox are all strings that we have formatted in a way so they appear tabulated. So, how would we do to fetch the corresponding row and column in your matrices in the SeatManager class? You can of course extract the row and col from the text by writing an algorithm, but that would be both risky and not always efficient.

A good solution might be to write an algorithm that converts the value of the SelectedIndex property to a corresponding pair of row and column numbers as illustrated in the figure above, where the item number 7 (counted from 0 at the top) in the ListBox corresponds to row = 0, and column 7 in the matrices in the SeatManager object.


9.4 To construct an algorithm for such a mapping, study the figure here at the right which demonstrates how an index in a one-dimensional array can be mapped to a pair of row and column indices in a two-dimensional array, by rolling out a two-dimensional array to a one-dimensional list! A value saved for example in the position (3, 1) has the position 10 in the one-dimensional array.

9.5 Now, let's try to figure out a little algorithm to change the index from a matrix system to a one-dimensional array. Assume that an item in the ListBox is selected by the user, and we must find out in which position (row, col) we should save the name and price for the reservation.

9.6 Examine the code snippets below:

```
''' <summary>
''' Determine an index for an element at (row, col) corresponding to a position
''' in a one-dimensional presentation of the matrix. Think of the matrix as being
''' rolled out into a one-dimensional array. In a 3 x 3 matrix, the element in
''' position (1,1) has an index 4 in one-dimensional array.
'''    20  11  22
'''    33  41  55
'''    60  7  99    Consider value(1,1) = 41
''' The above matrix can now be represented as one dimensional array. This makes
''' it easier to update the listbox in the GUI.
'''    20  11  22  33  41  55  60  7  99    value(4) = 41
'''    Index 0  1  2  3  4  5  6  7  8
'''
''' Hence, index (1,1) in the matrix corresponds to row 4 in the listbox (one-dim array).
''' </summary>
''' <param name="row">row number.</param>
''' <param name="col">column number</param>
''' <returns>The new index as explained above.</returns>
''' <remarks></remarks>
```

```
Public Function MatrixIndexToVectorIndex(ByVal row As Integer, ByVal col As Integer) As Integer
    Dim index As Integer = row * m_totNumOfCols + col
    Return index
End Function
```



9.7 To do the opposite in case we would need that, the code below shows the solution:


```
''' <summary>
''' Determines the index in the matrix (row, col) that corresponds to a given
''' index in a one-dim array (listbox). This method actually is a reverse process of
''' the method MatrixIndexToVectorIndex (see above). The parameter row contains
''' the input, i.e. index of the element in a one-dim array. The results (row and col)
''' are saved and sent back to the caller via the ref variables row, col.
''' </summary>
''' <param name="row">Input and output parameter.</param>
''' <param name="col">Output parameter.</param>
''' <remarks></remarks>
Public Sub VectorIndexToMatrixIndex(ByRef row As Integer, ByRef col As Integer)
    Dim vectorRow As Integer = row 'row in a vector
    row = Math.Ceiling((vectorRow / m_totNumOfCols)) 'row in a maxtrix
    col = vectorRow Mod m_totNumOfCols
End Sub
```



10. Multidimensional arrays vs one-dimensional array of objects

As a programmer you can almost never avoid lists of data of same type. Databases are of course one of the best solutions, but it is not always you work directly with databases. In scientific formulas, there occur data in form of tables representing some type of data, and it is of course more natural to program the tables as they are. Using two or multidimensional arrays has a big disadvantage, and that is readability of the code. Each dimension represents something that is not directly readable from the code. Results(5,6) does not reveal what the rows and columns represent as they don't have a name. It could be Results(month, year), Results(matchnr, and score) or whatever. A good documentation might help a lot.

Consider a three-dimensional array, `seats(5, 12, 50)` which for example represents the seat number 12 in row 5 that costs 50. How informative is this, really? Not at all I would say. Consider now the construction in which we use an array of Seat objects (See also Version 3 later in this document).

We have now an array of objects indexed from zero to something, say 499. `m_seatList(10).m_row` gives you the row number for the seat saved in position 10 and `m_seatList(10).m_Category` gives the type of a seat. Other dimensions have readable names like row, col, price, etc, so you don't have to guess which dimension represents which information.

I'm quite sure that you agree with me that the this code speaks for itself. In addition to having many dimensions (count each member, row, col, price, etc of the **Seat** class as one dimension), It is quite easy to work with and maintain such a structure.

In most cases even an array with one dimension, can be constructed as an array of objects in which the object belongs to a class with only one field. This will give you a more maintainable structure.

As far as two- and multidimensional arrays are concerned, I have through my many years of developing engineering programs, experienced that using arrays with two or more dimensions are not very practical, from a developer's point of view. It is much easier to make use of one-dimensional arrays of objects.

My experience is that any type of a multi-dimensional data can be represented to a one-dimensional array, just by rolling them out into a one-dimensional array, just as explained earlier in this document. You can then write a little function that maps a multi-dimensional array index to a position in a one-dimensional array to access the same value.

To summarize. Avoid using arrays that are more than one-dimensional, unless there are special reasons for it. Make a class and use a one-dimensional array with elements of the class. We are going to work this way in the coming assignments.

As a rule of thumb, as soon as you have a two- or multi-dimensional list of data, create a class and declare a field for each dimension. Then use a list of objects instead, and that's why the next version, array of objects, is included in this assignment. Using a list of objects offers not only good readability, it provides a lot of flexibility. You can use different types for the dimensions, you can use variable names for each dimension, and you can easily put in more dimensions.



Version 3: Array of objects

This version is optional and is presented as an extra exercise. It is desired by the cinema owner to store more data about a seat,

- Seat category (Business Class, Economy Class, Handicap Seat, Staff Seat) etc).
- The phone number of a customer.
- The first name and the last name separately.
- Rows may have different number of seats.

As you may have noticed from the descriptions so far, each time we have to add and handle a new type of data for a seat, there comes a lot of coding, if we are determined to continue using one or two dimensional arrays. To create more arrays is definitely not an effective way.

A good solution is to define a class, **Seat**, and encapsulate every data desired about a seat, and write methods to handle all operations about a seat. In the **SeatManager** class, you can then declare a fixed-size array (as in the previous versions) with elements of the Seat class. Fixed-sized arrays are sometimes referred to as static arrays, meaning that the size of the array is determined at compile time. We will be utilizing mostly dynamic arrays of objects in the future. During the rest of this course as well as in advanced courses (and also in our daily programmer lives), we will be using collections, which are advanced list types but are easy to work with. .NET supports several collection types.

Create a new project in your solution and name it CBSVersion3. Copy the files from Version 2, or create new ones and then copy code after your needs.

11. The MainForm Class

Design your GUI using your own fantasy. Write the **Seat** and **SeatManager** classes first and then come back to the MainForm and write code so everything works the way expected.

12. The Seat Class

Create a new class **Seat** and complete it with fields for the above data and methods necessary to perform its tasks. Use Enums to group different types of constant data and make your class prepared for handling more information about a seat.

```
Public Class Seat
    'which row the seat belongs to
    Private m_row As Integer
    'which column the seat belongs to
    Private m_col As Integer

    'first name of the customer (should be a part of a new class Customer)
    Private m_firstName As String
    'last name of the customer (should be a part of a new class Customer)
    Private m_lastName As String

    'Seat category, Business, Economy, etc.
    Private m_category As SeatCategory
    'Seat status whether the seat is reserved, vacant, etc.
    Private m_status As SeatStatus

    'put more fields if you wish

    'continue with constructors
    'continue with properties and methods
End Class
```

```
Public Enum SeatCategory
    Business
    Economy
    Handicap
    Staff          'some seats are reserved for staff
    Window
    Isle
End Enum

Public Enum SeatStatus
    Reserved
    Vacant
    Unavailable    'seat not available due to reparations
End Enum
```

13. The SeatManager Class

Create a new class (not much of the previous version may be useful). **Important:** In the code, shown here, the array is created, but as you may recall from the lessons, there are always two steps with using arrays: (1) creating the array object (ref variable) and (2) creating each of the elements. The **m_seatList** in the code example is a reference variable to an array that is going to contain elements of the Seat type, i.e. Seat objects. The variable is initiated by the compiler to a value of **Nothing**.

Any reference to an element of the array that is not yet created will cause an exception and an abnormal termination of the program. You must create the elements of an array of objects as soon as you need to save an object in the array. You can either call a method that creates all the elements with default values or create and delete every element when needed. The former alternative is mentioned only because of using a fixed sized array; otherwise, the latter alternative is to prefer.

You can create an element with the keyword **New**

```
m_seatList(index) = New Seat()
```

You can delete an element by letting it point to **Nothing**.

```
m_seatList(index) = Nothing
```

To do:

Write the following methods:

- AddnewSeat(arguments)
- DeleteSeat(index)
- ChangeSeat(index, arguments)
- GetSeat(index)
- GetSeatInfo(index)

You may need to write other methods too.

Hopefully, this document has given you answers to many of your questions. Use the forums to discuss questions that still remain ambiguous in your thoughts.

Good Luck!

Farid Naisan, Course Responsible and Instructor

```
Public Class SeatManager
    ' The value of m_totNumOfSeats is determined by the
    ' client object (MainForm) at construction time
    Private ReadOnly m_totNumOfSeats As Integer
    Private m_seatList() As Seat

    'Constructor with the total number of seats as input parameter
    Public Sub New(ByVal totNumOfSeats As Integer)
        m_totNumOfSeats = totNumOfSeats
        m_seatList = New Seat(m_totNumOfSeats - 1) {} 'create the m_seatList array
    End Sub

End Class
```

```
Private Sub TestArrays()
    'Arrays - step2: Create each of the elements (where and when you need them)
    'Creates the first object. An example only.

    'first element created
    m_seatList(0) = New Seat()

    'delete the first element
    m_seatList(0) = Nothing

    'delete the whole array
    m_seatList = Nothing

    'resize the array
    m_seatList = New Seat(m_totNumOfSeats * 2 - 1) {}
End Sub
```